
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 12.13 which is part of the ‘IMS_34229_IWD_13wk35’ test suite in ATS ‘iwd-EUTRA-B2012-03_D13wk35’.

3. Corrections required
This section describes the TTCN changes required to ‘iwd-EUTRA-B2012-03_D13wk35’ release.
Change 1 – f_IMS_Response_ToHeaderFor100TryingRX
	Function name
	f_IMS_Response_ToHeaderFor100TryingRX

	Reason for change
	Function f_IMS_Response_ToHeaderRX is used to check responses sent by the UE. The function assumes that each response sent by the UE has to fill in the to tag if it was not present in the request sent by SS. However, a 100 Trying response can omit a to tag, see RFC3261, Section 8.2.6.2.

	Summary of change
	Introduced a variant of f_IMS_Response_ToHeaderRX called f_IMS_Response_ToHeaderFor100TryingRX that is tailor made for this purpose.

	TTCN module
	Common/IMS/IMS_CommonProcedures_Registration

	MCC160 Comment
	Accepted in principle but implemented as enhancement of existing function (see below)


Before change (we show existing function f_IMS_Response_ToHeaderRX here and how it differs from the new function. Note again that we do not propose to change the existing function but that we propose the introduction of an additional function)
    /*

   * @desc      generate template for To header depending on whether the previous request has had a to-tog or not
   * @param     p_RequestMessageHeader

   * @return    template (present) To

   * @status    APPROVED (IMS, LTE, LTE_A, LTE_IRAT)
   */ 
function f_IMS_Response_ToHeaderRX(template (value) MessageHeader p_RequestMessageHeader) return template (present) To

  {

    var template (omit) charstring v_ToTagFromRequest := f_MessageHeader_GetToTag(p_RequestMessageHeader);

    var template (value) SipUrl v_SipUrl := f_Addr_Union_GetSipUrl(p_RequestMessageHeader.toField.addressField);   // @sic R5s130453 change 1 sic@

    var template (present) charstring v_ExpectedToTag := ?;

    if (ispresent(v_ToTagFromRequest)) {

      v_ExpectedToTag := valueof(v_ToTagFromRequest);      /* @sic R5-132063  sic@ */

    }

    return cr_ToWithTag(v_SipUrl, v_ExpectedToTag);
  }
After change

    /*

   * @desc      generate template for To header for 100 Trying. Such a response does not necessarily have to bring a to tag
   * @param     p_RequestMessageHeader

   * @return    template (present) To

   */
function f_IMS_Response_ToHeaderFor100TryingRX(template (value) MessageHeader p_RequestMessageHeader) return template (present) To

  {

    var template (omit) charstring v_ToTagFromRequest := f_MessageHeader_GetToTag(p_RequestMessageHeader);

    var template (value) SipUrl v_SipUrl := f_Addr_Union_GetSipUrl(p_RequestMessageHeader.toField.addressField);   // @sic R5s130453 change 1 sic@

    var template (present) charstring v_ExpectedToTag := ?;

    if (ispresent(v_ToTagFromRequest)) {

      v_ExpectedToTag := valueof(v_ToTagFromRequest);    /* @sic R5-132063  sic@ */

      return cr_ToWithTag(v_SipUrl, v_ExpectedToTag);

    }

    else {

      return cr_ToDef(v_SipUrl);

    }
  }
MCC160 Implementation

	  function f_IMS_Response_ToHeaderRX(template (value) MessageHeader p_RequestMessageHeader,

                                     boolean p_Is100Response := false) return template (present) To

  { 

    var template (omit) charstring v_ToTagFromRequest := f_MessageHeader_GetToTag(p_RequestMessageHeader);

    var template (value) SipUrl v_SipUrl := f_Addr_Union_GetSipUrl(p_RequestMessageHeader.toField.addressField);
    var template (present) To v_To;

    if (ispresent(v_ToTagFromRequest)) {

      v_To := cr_ToWithTag(v_SipUrl, valueof(v_ToTagFromRequest));      /* @sic R5-132063  sic@ */

    } 

    else if (not p_Is100Response) { 

      v_To := cr_ToWithTag(v_SipUrl);

    } 

    else {

      v_To := cr_ToDef(v_SipUrl);
    }

    return v_To;

  }


Change 2 – f_IMS_InviteResponse_100_MessageHeaderRX
	Function name
	f_IMS_InviteResponse_100_MessageHeaderRX

	Reason for change
	Function f_IMS_Response_ToHeaderFor100TryingRX introduced in Change 1 needs to be used in the right place, i.e., the place where 100 Trying responses are checked.

	Summary of change
	One such place could be identified: function f_IMS_InviteResponse_100_MessageHeaderRX. In that function, the call to f_IMS_Response_ToHeaderRX is replaced by call of new function proposed in Change 1.

	TTCN module
	Common/IMS/IMS_CommonProcedures_CallControl

	MCC160 Comment
	Accepted in principle (see change 1)


Before change

  function f_IMS_InviteResponse_100_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest) return template (present) MessageHeader

  { /* NOTE: acc. to A.2.2 there is no Session-ID in this message */

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Response := cr_MessageHeader_Dummy;

    v_MessageHeader_Response.via             := v_MessageHeader_Invite.via;

    v_MessageHeader_Response.fromField       := f_IMS_Response_FromHeaderRX(v_MessageHeader_Invite);  /* @sic R5s130453  sic@ */

    v_MessageHeader_Response.toField         := f_IMS_Response_ToHeaderRX(v_MessageHeader_Invite);    /* @sic R5s130453  sic@ */  

    v_MessageHeader_Response.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Response.cSeq            := v_MessageHeader_Invite.cSeq;

    return v_MessageHeader_Response;

  }
After change

  function f_IMS_InviteResponse_100_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest) return template (present) MessageHeader

  { /* NOTE: acc. to A.2.2 there is no Session-ID in this message */

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Response := cr_MessageHeader_Dummy;

    v_MessageHeader_Response.via             := v_MessageHeader_Invite.via;

    v_MessageHeader_Response.fromField       := f_IMS_Response_FromHeaderRX(v_MessageHeader_Invite);  /* @sic R5s130453  sic@ */

    v_MessageHeader_Response.toField         := f_IMS_Response_ToHeaderFor100TryingRX(v_MessageHeader_Invite);    /* @sic R5s130453  sic@ */  

    v_MessageHeader_Response.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Response.cSeq            := v_MessageHeader_Invite.cSeq;

    return v_MessageHeader_Response;

  }
Change 3 – f_IMS_BuildSDP_AnnexC11_Step1or7
	Function name
	f_IMS_BuildSDP_AnnexC11_Step1or7

	Reason for change
	In the UPDATE request, TTCN sends a=curr:qos sendrecv none instead of a=curr:qos local sendrecv

	Summary of change
	Added the missing argument to template cs_SDP_Attribute_curr_quos

	TTCN module
	Common/IMS/IMS_SDP_Messages

	MCC160 Comment
	Accepted


Before change

  function f_IMS_BuildSDP_AnnexC11_Step1or7(charstring p_Step,

                                            template (value) SDP_attribute p_SDP_Attribute_curr_quos_remote := cs_SDP_Attribute_curr_quos(c_remote))

    runs on IMS_PTC return template (value) SDP_Message

  {

    var boolean v_IsStep7 := (p_Step == "step 7");

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var charstring v_IPAddrOrigin := f_IMS_PTC_NW_Address_GetAddrStr();

    var charstring v_FmtAudio := tsc_SDP_FmtAudio;

    var charstring v_SessionVersion := tsc_SDP_SessionVersion1;

    var template (value) SDP_media_desc v_SDP_Media_Desc;

    var template (omit) SDP_attribute_list v_SDP_Attribute_List := {

      // media attributes:

      cs_SDP_Attribute_rtpmap(v_FmtAudio, tsc_CODEC_AMR_8000_1),

      cs_SDP_Attribute_fmtp(v_FmtAudio, tsc_AMR_Fmtp_DefaultParameters),

      cs_SDP_Attribute_ptime,

      cs_SDP_Attribute_maxptime

    };

    v_SDP_Attribute_List := f_SDP_Attributes_Add_TX(v_SDP_Attribute_List, cs_SDP_Attribute_sendrecv, v_IsStep7);

    // precondition attributes:

    v_SDP_Attribute_List := f_SDP_Attributes_Add_TX(v_SDP_Attribute_List, cs_SDP_Attribute_curr_quos(c_local), not v_IsStep7);         // @sic R5s130510 change 4 sic@

    v_SDP_Attribute_List := f_SDP_Attributes_Add_TX(v_SDP_Attribute_List, cs_SDP_Attribute_curr_quos(c_sendrecv), v_IsStep7);          // @sic R5s130510 change 4 sic@

    v_SDP_Attribute_List := f_SDP_Attributes_Add_TX(v_SDP_Attribute_List, p_SDP_Attribute_curr_quos_remote);

    v_SDP_Attribute_List := f_SDP_Attributes_Add_TX(v_SDP_Attribute_List, cs_SDP_Attribute_des_quos(c_mandatory, c_local));

    v_SDP_Attribute_List := f_SDP_Attributes_Add_TX(v_SDP_Attribute_List, cs_SDP_Attribute_des_quos(c_optional, c_remote));

    if (v_IsStep7) {v_SessionVersion := tsc_SDP_SessionVersion2; }

    v_SDP_Media_Desc := cs_SDP_Media_Desc(cs_SDP_Media_Audio(v_FmtAudio),

                                          cs_SDP_Bandwidth_List_Media(30, 0, 0),  // @sic R5-130751 sic@

                                          v_SDP_Attribute_List);

    return cs_SDP_Message_Def(cs_SDP_Origin(-, v_SessionVersion, v_AddrType, v_IPAddrOrigin), cs_SDP_Connection(v_AddrType, v_IPAddrOrigin), -, v_SDP_Media_Desc);

  }
After change

  function f_IMS_BuildSDP_AnnexC11_Step1or7(charstring p_Step,

                                            template (value) SDP_attribute p_SDP_Attribute_curr_quos_remote := cs_SDP_Attribute_curr_quos(c_remote))

    runs on IMS_PTC return template (value) SDP_Message

  {

    var boolean v_IsStep7 := (p_Step == "step 7");

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var charstring v_IPAddrOrigin := f_IMS_PTC_NW_Address_GetAddrStr();

    var charstring v_FmtAudio := tsc_SDP_FmtAudio;

    var charstring v_SessionVersion := tsc_SDP_SessionVersion1;

    var template (value) SDP_media_desc v_SDP_Media_Desc;

    var template (omit) SDP_attribute_list v_SDP_Attribute_List := {

      // media attributes:

      cs_SDP_Attribute_rtpmap(v_FmtAudio, tsc_CODEC_AMR_8000_1),

      cs_SDP_Attribute_fmtp(v_FmtAudio, tsc_AMR_Fmtp_DefaultParameters),

      cs_SDP_Attribute_ptime,

      cs_SDP_Attribute_maxptime

    };

    v_SDP_Attribute_List := f_SDP_Attributes_Add_TX(v_SDP_Attribute_List, cs_SDP_Attribute_sendrecv, v_IsStep7);

    // precondition attributes:

    v_SDP_Attribute_List := f_SDP_Attributes_Add_TX(v_SDP_Attribute_List, cs_SDP_Attribute_curr_quos(c_local), not v_IsStep7);         // @sic R5s130510 change 4 sic@

    v_SDP_Attribute_List := f_SDP_Attributes_Add_TX(v_SDP_Attribute_List, cs_SDP_Attribute_curr_quos(c_local, c_sendrecv), v_IsStep7);   

    v_SDP_Attribute_List := f_SDP_Attributes_Add_TX(v_SDP_Attribute_List, p_SDP_Attribute_curr_quos_remote);

    v_SDP_Attribute_List := f_SDP_Attributes_Add_TX(v_SDP_Attribute_List, cs_SDP_Attribute_des_quos(c_mandatory, c_local));

    v_SDP_Attribute_List := f_SDP_Attributes_Add_TX(v_SDP_Attribute_List, cs_SDP_Attribute_des_quos(c_optional, c_remote));

    if (v_IsStep7) {v_SessionVersion := tsc_SDP_SessionVersion2; }

    v_SDP_Media_Desc := cs_SDP_Media_Desc(cs_SDP_Media_Audio(v_FmtAudio),

                                          cs_SDP_Bandwidth_List_Media(30, 0, 0),  // @sic R5-130751 sic@

                                          v_SDP_Attribute_List);

    return cs_SDP_Message_Def(cs_SDP_Origin(-, v_SessionVersion, v_AddrType, v_IPAddrOrigin), cs_SDP_Connection(v_AddrType, v_IPAddrOrigin), -, v_SDP_Media_Desc);

  }
Change 4 – f_IMS_MTCallSetup_Common_Steps7_11
	Function name
	f_IMS_MTCallSetup_Common_Steps7_11

	Reason for change
	200OK response for UPDATE request brings an SDP body. So, Content-Type header must be allowed

	Summary of change
	Added content type application/sdp to receive template

	TTCN module
	Common/IMS/IMS_CommonProcedures_CallControl

	MCC160 Comment
	Accepted
Further changes:

· new template cr_ContentTypSDP := cr_ContentTypeDef("application/sdp");
· new template cs_ContentTypSDP := cs_ContentTypeDef("application/sdp");


Before change

  function f_IMS_MTCallSetup_Common_Steps7_11(template (value) INVITE_Request p_InviteRequest,

                                              template (value) SDP_Message p_SDP_MessageStep7,

                                              template (present) SDP_Message p_SDP_MessageStep8,

                                              InternetProtocol_Type p_ProtocolUsedForInvite,

                                              template MessageBody p_MessageBodyRinging := omit,

                                              float p_WaitDuration_Ringing := 5.0) runs on IMS_PTC return template (omit) MessageHeader

  {

    var IMS_DATA_RSP v_IMS_DATA_RSP;

    var template (value) MessageHeader v_MessageHeader_Update;

    var MessageHeader v_MessageHeader_Ringing;

    var InternetProtocol_Type v_ProtocolForUpdate := f_IMS_PTC_ImsInfo_GetTransportProtocol();

    var SipUrl v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrl();

    var boolean v_WaitingFor180 := true;

    var boolean v_WaitingFor200 := true;

    timer t_Timer := p_WaitDuration_Ringing;

    // Send UPDATE

    v_MessageHeader_Update := f_IMS_UpdateRequest_MessageHeaderTX(p_InviteRequest);

    IMS_Client.send(cas_IMS_Update_Request(cs_IMS_RoutingInfo(v_ProtocolForUpdate), cs_UPDATE_Request(v_ContactUrl, v_MessageHeader_Update, cs_MessageBody_SDP(p_SDP_MessageStep7))));

    t_Timer.start; // @sic R5-125772 sic@ Have introduced the timer to accommodate optional messages that may not arrive

    while (v_WaitingFor180 or v_WaitingFor200) {

      alt {

        // Step 8. Receive 200 OK. The UE acknowledges the UPDATE with 200 OK and includes SDP answer to acknowledge its current precondition status.

        [v_WaitingFor200] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_ProtocolForUpdate),

                                                              cr_Response(c_statusLine200, f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Update), cr_MessageBody_SDP(p_SDP_MessageStep8))))

          -> value v_IMS_DATA_RSP

          {

            // check at least one connections field sent

            f_IMS_SDP_CheckConnections(v_IMS_DATA_RSP.Response.messageBody.sdpMessageBody);

            v_WaitingFor200 := false;

          }

        [v_WaitingFor180] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(p_ProtocolUsedForInvite),

                                                              cr_Response(c_statusLine180, f_IMS_InviteResponse_180_MessageHeaderRX(p_InviteRequest), p_MessageBodyRinging)))

          -> value v_IMS_DATA_RSP

          {

            t_Timer.stop;

            v_MessageHeader_Ringing := v_IMS_DATA_RSP.Response.msgHeader;

            v_WaitingFor180 := false;

          }

        [] t_Timer.timeout

          {

            v_WaitingFor180 := false;

          }

      }

    }

    if (isbound(v_MessageHeader_Ringing)) {

      f_IMS_MessageHeader_CheckContactAddr(v_MessageHeader_Ringing);      // check whether contact contains "SIP URI with IP address or FQDN" (see @sic R5s120727 sic@)

      if (f_MessageHeader_CheckRequire(v_MessageHeader_Ringing, c_tag100rel)) {

        // Step 10 - 11. Send PRACK, Receive 200 OK if the 180 response contains 100rel option tag within the Require header

        f_IMS_MTCallSetup_SendPRACK_ReceiveOK(p_InviteRequest, v_MessageHeader_Ringing);

      }

      return v_MessageHeader_Ringing;

    }

    return omit;

  }
After change

  function f_IMS_MTCallSetup_Common_Steps7_11(template (value) INVITE_Request p_InviteRequest,

                                              template (value) SDP_Message p_SDP_MessageStep7,

                                              template (present) SDP_Message p_SDP_MessageStep8,

                                              InternetProtocol_Type p_ProtocolUsedForInvite,

                                              template MessageBody p_MessageBodyRinging := omit,

                                              float p_WaitDuration_Ringing := 5.0) runs on IMS_PTC return template (omit) MessageHeader

  {

    var IMS_DATA_RSP v_IMS_DATA_RSP;

    var template (value) MessageHeader v_MessageHeader_Update;

    var MessageHeader v_MessageHeader_Ringing;

    var InternetProtocol_Type v_ProtocolForUpdate := f_IMS_PTC_ImsInfo_GetTransportProtocol();

    var SipUrl v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrl();

    var boolean v_WaitingFor180 := true;

    var boolean v_WaitingFor200 := true;

    timer t_Timer := p_WaitDuration_Ringing;

    // Send UPDATE

    v_MessageHeader_Update := f_IMS_UpdateRequest_MessageHeaderTX(p_InviteRequest);

    IMS_Client.send(cas_IMS_Update_Request(cs_IMS_RoutingInfo(v_ProtocolForUpdate), cs_UPDATE_Request(v_ContactUrl, v_MessageHeader_Update, cs_MessageBody_SDP(p_SDP_MessageStep7))));

    t_Timer.start; // @sic R5-125772 sic@ Have introduced the timer to accommodate optional messages that may not arrive

    while (v_WaitingFor180 or v_WaitingFor200) {

      alt {

        // Step 8. Receive 200 OK. The UE acknowledges the UPDATE with 200 OK and includes SDP answer to acknowledge its current precondition status.

        [v_WaitingFor200] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_ProtocolForUpdate),

                                                              cr_Response(c_statusLine200, f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Update,cs_ContentTypeDef("application/sdp")), cr_MessageBody_SDP(p_SDP_MessageStep8)))) 

          -> value v_IMS_DATA_RSP

          {

            // check at least one connections field sent

            f_IMS_SDP_CheckConnections(v_IMS_DATA_RSP.Response.messageBody.sdpMessageBody);

            v_WaitingFor200 := false;

          }

        [v_WaitingFor180] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(p_ProtocolUsedForInvite),

                                                              cr_Response(c_statusLine180, f_IMS_InviteResponse_180_MessageHeaderRX(p_InviteRequest), p_MessageBodyRinging)))

          -> value v_IMS_DATA_RSP

          {

            t_Timer.stop;

            v_MessageHeader_Ringing := v_IMS_DATA_RSP.Response.msgHeader;

            v_WaitingFor180 := false;

          }

        [] t_Timer.timeout

          {

            v_WaitingFor180 := false;

          }

      }

    }

    if (isbound(v_MessageHeader_Ringing)) {

      f_IMS_MessageHeader_CheckContactAddr(v_MessageHeader_Ringing);      // check whether contact contains "SIP URI with IP address or FQDN" (see @sic R5s120727 sic@)

      if (f_MessageHeader_CheckRequire(v_MessageHeader_Ringing, c_tag100rel)) {

        // Step 10 - 11. Send PRACK, Receive 200 OK if the 180 response contains 100rel option tag within the Require header

        f_IMS_MTCallSetup_SendPRACK_ReceiveOK(p_InviteRequest, v_MessageHeader_Ringing);

      }

      return v_MessageHeader_Ringing;

    }

    return omit;

  }

